MODULE 1
Impurity atomic species in fusion plasma, their ionisation state and radiating characteristics - the ADAS approach.

Demonstration script

Hugh Summers, Martin O'Mullane and Alessandra Giunta

September 26, 2013

Contents

1 Demo (a) Finding about ADAS

```
------------------------------
DEMO A: Finding about ADAS
------------------------------
PURPOSE:Look at the ADAS, ADAS-EU and OPEN-ADAS websites and their structure.

EXAMPLE: The are three main websites related to ADAS:
1. http://www.adas.ac.uk: this website contains all information about ADAS, its
history, news.  It includes the online manual, which gives an overview of the
theory behind and detailed explanation of all  ADAS series and data formats. All
material related to dissemination (e.g. bullettins, publications, theses) is
available in the website, together with information about past and future ADAS
courses and workshops. Links to ADAS-EU and OPEN-ADAS are also present.
2. http://www.adas-fusion.eu: this website is striclty related to the ADAS-EU
project in support of fusion laboratories in Europe and for ITER (Euratom Framework
7 Support Action). It provides all information about the work packages supporting
fusion research and implemented by ADAS. It also gives details on ADAS-EU courses
and links to ADAS and OPEN-ADAS websites.
3. http://open.adas.ac.uk: this website enables non members to download and use
ADAS data. It allows one multiple research choice:
- a freedom research (e.g. typing the name of the elements or ion or wavelenth
range of interest);
- a research according to the wavelength of interest
- a research based on ion selecting the element from the periodic table.


DEMO a1: The ADAS and ADAS-EU websites
ADAS wesite
1. Open an internet browser and type http://www.adas.ac.uk.
2. Click on "Manual"
3. Look at the different ADAS series (ADAS102, ADAS102, etc.)
4. Scrolling down, look at the different ADAS data formats (adf00, adf01, etc.)
(Example files: demo_a1_adas_web.png, demo_a1_adas_web_series.png,
demo_a1_adas_web_adf.png)

ADAS-EU website
5. Click on "ADAS-EU" and, in the webpage which appears, click to the link
http://www.adas-fusion.eu. Alternatively, type directly on the browser
http://www.adas-fusion.eu.
6. Look at the ADAS-EU website.
(Example files: demo_a1_adaseu_web.png)


DEMO a2: OPEN-ADAS
1. Type on the browser http://open.adas.ac.uk.
2. Click on "Freedom" to look at the freedon research.
3. Then click on "Ion" and choose an element from the periodic table.
(Example files: demo_a2_open_adas_search.png, demo_a2_open_adas_z.png)
```

1.1 Demo (a) Figures

1.1.1 Demo (a-1) demo_a/demo_a1_adaseu_web.png

Figure 1: ADAS-EU website - first page

1.1.2 Demo (a-1) demo_a/demo_a1_adas_web.png



Figure 2: ADAS website - first page

1.1.3 Demo (a-1) demo_a/demo_a1_adas_web_series.png



Figure 3: ADAS - code series

1.1.4 Demo (a-1) demo_a/demo_a1_adas_web_adf.png



Figure 4: ADAS data formats

1.1.5 Demo (a-2 demo_a/demo_a2_open_adas_search.png



Figure 5: OPEN-ADAS - freeform search

1.1.6 Demo (a-2 demo_a/demo_a2_open_adas_z.png



Figure 6: OPEN-ADAS - search by ion

2 Demo (b) The code and data organisation

```
------------------------------
DEMO B: The code and data organisation
------------------------------
PURPOSE: Overview of ADAS database and code organisation. This demo shows the
interactive and non-interactive capabilities of ADAS. It also illustrates how to
access the atomic and molecular database and look at the different libraries and
codes (fortran,idl, etc.).

EXAMPLE: ADAS provides atomic data as well as sets of codes for modelling the
radiating properties of ions and atoms in plasmas and for assisting in the
interpretation and analysis of spectral measurements. It is addressed to a large
variety of plasmas, ranging from astrophysical (from solar atmosphere to
interstellar medium) to laboratory devices (from fusion to technological plasmas).

Documentations:  The entire ADAS manual and related documentation are in the
directory /home/adas/doc/

Componemts of ADAS structure:
1. an extensive database of fundamental and derived atomic data. They are collected
in the directory /home/adas/adas/

adf00  adf06  adf12  adf18  adf24  adf34  adf48    arch103  arch603    scripts409
adf01  adf07  adf13  adf19  adf25  adf35  adf49    arch105  arch804    scripts416
adf02  adf08  adf14  adf20  adf26  adf38  adf54    arch106  mdf00
adf03  adf09  adf15  adf21  adf27  adf39  adf56    arch108  mdf02
adf04  adf10  adf16  adf22  adf28  adf40  arch101  arch601  scripts405
adf05  adf11  adf17  adf23  adf32  adf42  arch102  arch602  scripts406


adf     is ADAS data file
mdf     is molecular data format
arch    are archives
scripts are drivers

2. a large set of libraries, routines and utilities (including FORTRAN, C,C++, IDL
and MATLAB) for accessing the database, delivering data, performing calculation of
fundamental data and spectroscopic analysis (ADAS series, e.g. ADAS201).
Some of the most relevent directories are the following:

- /home/adas/fortran/
which contains:

adas1xx  adas3xx  adas5xx  adas7xx  adas9xx
adas2xx  adas4xx  adas6xx  adas8xx  adaslib

-/home/adas/C/
which contains:
```

```
adas5xx  adas6xx  adaslib

-/home/adas/idl/
which contains:

adas1xx  adas3xx  adas5xx  adas7xx  adas9xx  write_adf
adas2xx  adas4xx  adas6xx  adas8xx  adaslib  read_adf

The subdirectories adaslib, write_adf and read_adf include utility and reading
routines.

Each of the subdirectories of the type adas1xx,adas2xx etc are related to the ADAS
series:  e.g. adas2xx is ADAS series 2 for the investigation of "General Z data and
Population Processing". In turn this includes the directories:

adas201  adas203  adas205  adas207  adas209  adas211  adas213  adas215  adaslib
adas202  adas204  adas206  adas208  adas210  adas212  adas214  adas216

for specific displaying or analysis purposes: e.g. adas201 contains a set of
routines which allows one to graph and fit coefficients for specific Z excitation.

Capabilities of ADAS structure:
1. the interactive part, which provides immediate display of fundamental and
derived data used in analysis and allow one to explore parameter dependencies and
diagnostic predictions of atomic population and plasma models;
2. the non-interactive part, provides a set of subroutines than can be accessed
from the users  and embedded in their own codes.


COMMENTS: The use of ADAS and its full capabilities, as described by this demo,  is
reserved to the members. However, the database (fundamental and derived atomic
data) and the fortran reading routines for the ADAS data files are available also
in the OPEN-ADAS website (see demo a) and free from the restriction of ADAS
membership..


DEMO b1: Language and library structure
1. Open the directory /home/adas/doc/ to look at the documentation.
2. Open the directory /home/adas/adas/ to look at the location of the atomic and
molecular database.
3. Open the directory /home/adas/idl/ to look at the ADAS series.
4. Open the directory /home/adas/fortran/ to look at the ADAS series (as point 3.).

DEMO b2: ADAS data formats
1. Open the directory /home/adas/adas/adf04 to look at an example of ADAS data
file.

DEMO b3: Interactive ADAS menus and starting code
1. Go into the directory /home/adas/pass/.
2. Run adas.
```

3. Click on series 2: "2 General Z Data and Population Processing".
4. Click on "ADAS201: Specific Z Excitation File - Graph and Fit Coefficient"
5. Select a data set, e.g. from /home/adas/adas/adf04/adas#6/mom97_ls#c1.dat and start the program.

3 Demo (c) Fundamental data collections

```
------------------------------
DEMO C: Fundamental data collections
------------------------------
PURPOSE:Look at the fundamental data collection.

EXAMPLE: This demo gives examples of how look at some of the fundamental data
collected in the ADAS data files (adf).
Three samples are provided:

1. adf04: These ADAS data files are collections of energy levels (or terms
or configuration arrays according to the resolution adopted), radiative
transitions probabilities (A-values) and rate coefficients for specifed low
states of an ion. There are five types of adf04: a) Type 1: electron
collision transition line Omega (collision strength) as a function of the
threshold parameter X. X, with 1<X <infinitive, is defined as ei/(Ej-Ei),
where ei is the incident  electron energy, Ej the energy of the upper state
of the transition and Ei the energy of the  lower state (Note that
ei+Ei=ej+Ej, where ej is the energy of the scattered electron).  b) Type 3:
electron collision transition line Upsilon (effective collision strength) as
a function of electron temperature. c) Type 4: electron collision transition
line Upsilon and Downsilon as a function of energy (see Bryans 2005 -
http://www.adas.ac.uk/theses/bryans_thesis.pdf). d) Type 5: electron
collision transition line Omega as a function of the final state energy
(which is the enery of scattered electron ej). e) Type 6: electron collision
transition line Omega_l as a function of l.

For the demo a distorted wave adf04 for Be-like oxygen (O+4) has been
chosen: /home/adas/adas/adf04/cophps#be/dw/ls#o4_t5.dat

This is a type 5 adf04 in ls resolution. The term configurations are in the
Eissner notation.

Three different transitions, which involve the ground term, have been
chosen: 1) a dipole transition: 2s2 1S - 2s 3p 1P (corresponding to indices
1-9) 2) a spin change transition: 2s2 1S - 2s 3p 3P (corresponding to
indices 1-10) 3) a monopole transition: 2s2 1S - 2p 3p 1S (corresponding to
indices 1-24)

The three transitions show three characteristic behaviour (see viewgraphs).

2. adf08: This data file set contains the radiative recombination
coefficients. For ionised H which recombines to neutral H, the adf08 is:
/home/adas/adas/adf08/rrc98##/rrc98##_h1.dat. This is in LS resolution.
Consider the first three shells n=2,3,4. For n=2 shell the corresponding
terms are 2s 2S and 2p 2P, with indices 2 and 3 respectively. For n=3 shell
the corresponding terms are 3s 2S, 3p 2P and 3d 2D with indices  4, 5 and 6
respectively. Finally, for n=4 shell the corresponding terms are 4s 2S, 4p
2P, 4d 2D and 4f 2F with indices 7, 8, 9 and 10 respectively. The
```

contribution of radiative recombination to each shell is given by the sum of the term contributions.

3. adf09: This data file provides final state level-resolved dielectronic recombination rate coefficients into final terms (LS resolution) or levels (J-resolved). For Be-like -> B-like carbon (C+2 -> C+1), in LS resolution and considering the recombination from n=2 shell of recombining ion into n=2 shell of recombined ion, the adf09 is:
/home/adas/adas/adf09/nrbjc00#be/nrb00#be_c2ls22.dat The total contribution due to dielectronic recombination is calculated for each metastable of the recombining ion. In this case the recombining ion is C+2 and the metastable terms are 2s2 1S and 2s 2p 3P (see also MODULE 2 DEMO a: Identifying metastables).

COMMENTS: Note that an adf04 type 5 has been used in DEMO 1c instead of an adf04 type 1. Therefore the final energy ej (energy of scattered electron) has been converted into the X parameter for each transitions, which is X=(ej+delta_E)/delta_E, where delta_E is the equals to Ej-Ei, that is the energy difference between level j and level i of the corresponding transition.

DEMO c1: Looking at adf04: understanding the cross sections
1. Look at the adf04 type 5 for O+4 and select three transitions: e.g. dipole, spin change and monopole.
2. Use read_adf04.pro to read the adf04 selected.
3. Convert type 5 in type 1 adf04 to show the behaviour of the collision strength of each transition as a function of the threshold parameter.
4. Plot the Omega for the three transitions.
Program: demo_c_1.pro
Sample of output file: demo_c_1.ps

DEMO c2: Looking at adf08: radiative recombination data.
1. Look at the adf08 for H to select the appropriate level indices.
2. Use read_adf08.pro to read the selected adf08.
3. Sum over the shell (n=2, n=3, n=4).
4. Plot the radiative recombination coefficients into n=2, n=3 and n=4 shells for H as a function of electron temperature.
Program: demo_c_2.pro
Sample of output file: demo_c_2.ps

DEMO c3: Looking at adf09: dielectronic recombination data
1. Use xxdata_09.pro to read adf09.
2. Plot the total contribution of dielectronic recombination from the two metastables of recombining ion.
Program: demo_c_3.pro
Sample of output file: demo_c_3.ps

3.1 Demo (c) Figures

3.1.1 Demo (c-1) demo_c/demo_c_1.pdf



Figure 7:

3.1.2 Demo (c-2) demo_c/demo_c_2.pdf



Figure 8:

3.1.3 Demo (c-3) demo_c/demo_c_3.pdf



Figure 9:

3.2 Demo (c) Procedures

3.2.1 Demo (c-1) demo_c/demo_c_1.pro

```
pro demo_c_1
;Use read_adf04.pro to read type 5 adf04 for O+4 (Be-like)
;select three different transitions (e.g. dipole, spin change and monopole)
;plot the omega as a function of the threshold parameter [X=ei/(Ej-Ei)]

;adf04 type 5 in ls resolution
adf04='/home/agiunta/adas/adf04/cophps#be/dw/ls#o4_t5.dat'

;read the adf04 using read_adf04.pro
read_adf04,file=adf04,fulldata=data

;dipole transition: e.g. 2s2 1S - 2s 3p 1P
dd=where(data.lower eq 1 and data.upper eq 9)

;spin exchange transition: e.g. 2s2 1S - 2s 3p 3P
ee=where(data.lower eq 1 and data.upper eq 10)

;monopole transition: e.g. 2s2 1S - 2p 3p 1S
mm=where(data.lower eq 1 and data.upper eq 24);8)

ej=data.te

;Derive the X parameter for the three transitions

;1. Dipole transition
;identify the levels
d1=where(data.ia eq 1)
d2=where(data.ia eq 9)
;Values of energy in cm-1 corresponding to the previous levels
Ej_d=data.wa[d2]
Ei_d=data.wa[d1]
;calculate the delta_E for the dipole transition and convert to eV
delta_E_d=(Ej_d-Ei_d)/8066.0
;convert eV to Ryd
delta_E_d=delta_E_d/13.6
X_d=dblarr(n_elements(ej))
for i=0,n_elements(ej)-1 do X_d[i]=(ej[i]+delta_E_d)/delta_E_d


;2. Spin exchange transition
;identify the levels
s1=where(data.ia eq 1)
s2=where(data.ia eq 10)
;Values of energy in cm-1 corresponding to the previous levels
Ej_s=data.wa[s2]
Ei_s=data.wa[s1]
```

```
;calculate the delta_E for the dipole transition and convert to eV
delta_E_s=(Ej_s-Ei_s)/8066.0
;convert eV to Ryd
delta_E_s=delta_E_s/13.6
X_s=dblarr(n_elements(ej))
for i=0,n_elements(ej)-1 do X_s[i]=(ej[i]+delta_E_s)/delta_E_s


;3. Monopole transition
;identify the levels
m1=where(data.ia eq 1)
m2=where(data.ia eq 24)
;Values of energy in cm-1 corresponding to the previous levels
Ej_m=data.wa[m2]
Ei_m=data.wa[m1]
;calculate the delta_E for the dipole transition and convert to eV
delta_E_m=(Ej_m-Ei_m)/8066.0
;convert eV to Ryd
delta_E_m=delta_E_m/13.6
X_m=dblarr(n_elements(ej))
for i=0,n_elements(ej)-1 do X_m[i]=(ej[i]+delta_E_m)/delta_E_m


;plot Omega as a function of the threshold parameter X
set_plot,'ps'
device, /isolatin1, font_index=8
device, bits=8, filename='demo_c_1.ps',  $
        font_size = 14, xsize=18.0, ysize=16.0, $
        yoffset=7.0, /color
device, /helvetica

;using a normalisation factor to show the behaviour of the three transitions
plot_oi,X_d,data.gamma[dd,*],xrange=[1.,10.],yrange=[0.,0.15],$
        title='O+4',xtitle='X',ytitle='!7X!3'
plots,X_s,data.gamma[ee,*],line=1
plots,X_m,data.gamma[mm,*],line=2

xyouts,1.8,0.1,'dipole'
xyouts,5.,0.007,'spin change'
xyouts,2.,0.023,'non-dipole'

device, /close
set_plot,'X'
!p.font=-1

end
```

3.2.2 Demo (c-2) demo_c/demo_c_2.pro

```
pro demo_c_2
```

```
;Use read_adf08 to read adf08 for H
;and extract and plot state selective radiative
;recombination coefficients for H into n=2, n=3 and n=4 shells


adf08='/home/adas/adas/adf08/rrc98##/rrc98##_h1.dat'
parent=1
;define the level indices looking at the adf08 selected
level=[2,3,4,5,6,7,8,9,10]
nlev=n_elements(level)

;define electron temperature (eV)
te=adas_vector(low=0.1,high=1.e5,num=30)
nte=n_elements(te)

rec=fltarr(nte,nlev)

;read the adf08 selected using read_adf08.pro
for i=0,nlev-1 do begin
  read_adf08,file=adf08,parent=parent,level=level[i],te=te,data=data
  rec[*,i]=data
endfor

;sum over the shell
;n=2: level=2 -> 2s 2S; level+3 -> 2p 2P
recn2=total(rec[*,0:1],2)

;n=3: level=4 -> 3s 2S;level=5 -> 3p 2P;level=6 -> 2d 2D
recn3=total(rec[*,2:4],2)

;n=4: level=7 -> 4s 2S;level=8 -> 4p 2P;level=9 -> 4d 2D; level=10 -> 4f 2F
recn4=total(rec[*,5:8],2)

;plot recombination coefficients to n=2, n=3 and n=4
set_plot,'ps'
device, /isolatin1, font_index=8
device, bits=8, filename='demo_c_2.ps',  $
        font_size = 14, xsize=18.0, ysize=16.0, $
        yoffset=7.0, /color
device, /helvetica

plot_oo,te,recn2,title='H!u+1!n + e-> H!u+0!n + h!7m!3',$
                    xtitle='ELECTRON TEMPERATURE (eV)', $
                    ytitle='RECOMBINATION COEFFICIENT (cm!u3!n s!u-1!n)'

oplot,te,recn3
oplot,te,recn4

xyouts,te[nte/5],recn2(nte/5),'n=2'
xyouts,te[nte/3],recn3(nte/3),'n=3'
```

```
xyouts,te[nte/2],recn4(nte/2),'n=4'


device, /close
set_plot,'X'
!p.font=-1

end
```

3.2.3 Demo (c-3) demo_c/demo_c_3.pro

```
pro demo_c_3
;Use xxdata_09.pro to read adf09 for C Be-like -> B-like
;in ls resolution from n=2 -> n=2

adf09='/home/adas/adas/adf09/nrbjc00#be/nrb00#be_c2ls22.dat'

;read adf09 using use xxdata_09.pro
xxdata_09,file=adf09,fulldata=data

;convert electron temperature units from Kelvin to eV
te=data.tea/11604.5

;plot the total contribution of dielectronic recombination
;from the two metastables of recombining ion
set_plot,'ps'
device, /isolatin1, font_index=8
device, bits=8, filename='demo_c_3.ps',         $
        font_size = 14, xsize=18.0, ysize=16.0, $
        yoffset=7.0, /color
device, /helvetica
plot,alog10(te),alog10(data.diel_tot[0,*]),                          $
                yrange=[-20.,-10.],                                  $
                title='C Be-like -> B-like',                         $
                xtitle='log(T!de!n) (eV)',                           $
                ytitle='log(!7a!3!dtot!n!u(d)!n) (cm!u3!n s!u-1!n)'
oplot,alog10(te),alog10(data.diel_tot[1,*]),line=1

xyouts,0.9,-10.5,'1st metastable'
xyouts,-0.8,-13.,'2nd metastable'

device, /close
set_plot,'X'
!p.font=-1

end
```

4 Demo (d) Executing population calculations

```
------------------------------
DEMO D: Executing population calculations
------------------------------
PURPOSE: Perform population calculations, look at the different coefficients and
calculate ionisation balance and radiated power.

The population of excited levels is calculated by ADAS205 and ADAS208. The input
dataset is a specific ion file in the adf04 format. ADAS208 is an extension of
ADAS205 and includes in the population calculation: a) metastable parents;
b)condensed influenced of very highly excited level populations; c)inner shell
ionisation forming excited states; d) three boby recombination.
Additionally, ADAS208 generates the Generalised Collisional-Radiative (GCR)
coefficients:
1) adf15, which is the Photon Emissivity Coefficients, PEC;
2) adf11, which includes the following coefficients:

   scd= ionisation coefficients
   acd= free electron recombination coefficients
   ccd= charge exchange recombination coefficients
   xcd= metastable cross-coupling coefficients
   qcd= parent metastable cross-coupling coefficients

   plt= total excitation line power coefficients
   prb= free electron recombination cascade bremsstrahlung power coefficients
   prc= charge exchange power coefficients
   pls= specific line excitation power coefficients

   ecd= effective ionisation potential
   zcd= effective superstage charge
   ycd= effective superstage square of effective charge

   The units of first group are cm3 s-1, while for the power coefficients
   the units are ergs cm3 s-1. The last group is used for heavy species.

   The adf11 data files are identified by year (e.g. 85, 89, 93, 96).
   96 -> full GCR (also 93 is GCR, however they are available only for a small set
   of elements)

Once all the GCR coefficients adf11 are available (metastable resolved and stage to
stage), fractional abundances and radiated power may be calculated using ADAS405 at
equilibrium and ADAS406 using a simple transient model.

EXAMPLE: For the population calculation, this demo provides an example of the use
of ADAS205 applied to neutral helium. The population of excited levels is calculate
with respect to the ground or metastable level population.

The behaviour of the ratio between excited level population and ground level
population as a function of electron density helps to identify:
```

```
1) different plasma regimes: coronal, CR, LTE
2) metastable levels or terms (see also MODULE 2 DEMO a: Identifying
   metastables).

The adf04 selected for the demo is:
/home/adas/adas/adf04/adas#2/mom97_ls#he0.dat

The temperature chosen is 8.6 eV.

Two outputs from the advanced calculation performed by ADAS208 are included in the
demo:
1) adf15 for C+1 (excitation PEC):
   /home/adas/adas/adf15/pec96#c/pec96#c_pjr#c1.dat
   spectral line: 858.4 Ang.  2s2 2p 2P - 2s2 3s 2P
   C+1 has two metastable terms: 2s2 2p 2P, which is the ground term, and 2s 2p2 4P.
   The excitation PECs driven by the two metastables have been chosen for the demo.
2) adf11 for O+3 (free electron recombination coefficient, acd):
   In central ADAS the data file is:
   /home/adas/adas/adf11/acd96r/acd96r_o.dat
   However, if one wants to investigate the behaviour at high density, it would be
   better to increase the electron density range at which the coefficient is
   calculated. This is done running ADAS404 interactively for oxygen and editing
   the electron temperature and density range. The output file used as example has
   been saved locally and named acd404_o.pass.

   O+4 has 2 metastable terms, 2s2 1S and 2s 2p 3P, as well as O+3 which has the 2
   metastable terms 2s2 2p 2P and 2s 2p2 4P.
   The recombination coefficient from the second metastable of O+4 (iprt=2) to the
   second metastable of O+3 (igrd=2) has been selected for this demo.


The adf11 data files are of two types: stage to stage or unresolved (also called
standard type) and metastable resolved - e.g. the file
/home/adas/adas/adf11/acd96r/acd96r_o.dat, mentioned above provides metastable
resolved recombination coefficients, while the file
/home/adas/adas/adf11/acd96/acd96_o.dat gives stage to stage recombination
coefficients. Once the adf11 files (scd, acd, etc) are available for an element, it
is possible to calculate the fractional abundances and the radiated power at
equilibrium using ADAS405.

For oxygen the inputs are the following:

Element symbol= o
Year=96
Default year=96
Isonuclear master classes: scd, acd, prb, plt (for stage to stage)
                           scd, acd, qcd, xcd prb, plt (for metastable resolved)
```

COMMENTS: Note that the full GCR adf15 used for the demo is in LS resolution.

DEMO d1: Running ADAS205 for populations
1. Use ADAS205 with the interactive ADAS windows.
   (sample of output file: demo_d1.ps)

DEMO d2: Looking at adf15 and adf11
1. Look at the adf15 selected to identify blocks corresponding to the
   line chosen (C II 858.4 Ang.).
2. Use read_adf15.pro to read the metastable resolved PEC data file adf15
3. Plot a surface of the excitation PEC as a function of Te and Ne comparing the
behaviour due to the two metastables of C+1.
3. Look at the adf11 year 96 resolved for oxygen. Look at the electron
temperature and density range. If it is not appropriate, run the interactive
ADAS404, with a wider range of Te and Ne, to produce a new adf11 for oxygen and
save it locally.
4. Use read_adf11.pro to read the selected adf11.
5. Plot a surface of ACD as a function of electron temperature and density.
Program: demo_d_2.pro
Output files: demo_d_2_pec.ps, demo_d_2_acd.ps

DEMO d3: Running ADAS405 interactively and offline
1. Using the input defined above run ADAS405 using the interactive windows:
   a. Select -> Type of master files: Standard
   b. Select -> Data file: NULL
   c. Produce a fractional abundance plot and a power function plot
      (output files: demo_d_3_intv_ion_unres.ps, demo_d_3_intv_power_unres.ps)
   d. Repeat a.- c. selecting Type of master files: Partial - Resolved
      (output files: demo_d_3_intv_ion_res.ps, demo_d_3_intv_power_res.ps)
2. Use run_adas405.pro from the command line for oxygen.
3. Plot stage to stage fractional abundances as a function of electron temperature
   for a fixed electron density (e.g. dens=1.e10 cm-3).
4. Plot metastable resolved fractional abundances as a function of electron
   temperature for a fixed electron density (e.g. dens=1.e10 cm-3).
5. Plot metastable resolved radiated power (total, plt, prb and ion) as a function
   of electron temperature for a fixed electron density (e.g. dens=1.e10 cm-3).
   Program: demo_d_3.pro
   Output files: demo_d_3_ion_unres.ps, demo_d_3_ion_res.ps, demo_d_3_power_res.ps

4.1 Demo (d) Figures

4.1.1 Demo (d-1) demo_d/demo_d_1.pdf



Figure 10:

4.1.2 Demo (d-2) demo_d/demo_d_2_acd.pdf



Figure 11:

4.1.3 Demo (d-2) demo_d/demo_d_2_pec.pdf



Figure 12:

4.1.4 Demo (d-3) demo_d/demo_d_3_intv_ion_res.pdf



Figure 13:

4.1.5 Demo (d-3) demo_d/demo_d_3_intv_ion_unres.pdf



Figure 14:

4.1.6 Demo (d-3) demo_d/demo_d_3_intv_power_res.pdf



Figure 15:

4.1.7 Demo (d-3) demo_d/demo_d_3_intv_power_unres.pdf



Figure 16:

4.1.8 Demo (d-3) demo_d/demo_d_3_ion_res.pdf



Figure 17:

4.1.9 Demo (d-3) demo_d/demo_d_3_ion_unres.pdf



Figure 18:

4.1.10 Demo (d-3) demo_d/demo_d_3_power_res.pdf



Figure 19:

4.2 Demo (d) Procedures

4.2.1 Demo (d-3) demo_d/demo_d_3.pro

```
pro demo_d_3
;Use run_adas405.pro to produce stage to stage and metastable resolved
;fractional abundances and radiated power for oxygen

;Define electron temperature (eV) array
te=adas_vector(low=1.,high=1000.,num=24)

;Define a constant electron density (cm-3)
dens=1.e10

;Set up source (central ADAS or user directory),
;element, year and default year

uid='adas'
elem='o'
year=96
defyear=96

;Run run_adas405.pro to produce both fractional abundances and power
;in the unresolved or standard (stage to stage) case
run_adas405, uid=uid, year=year, defyear=defyear, elem=elem, te=te,$
             dens=dens, frac=frac_unres, power=power_unres,/all

;Run run_adas405.pro to produce both fractional abundances and power
;in the metastable resolved case
run_adas405, uid=uid, year=year, defyear=defyear, elem=elem, te=te,$
             dens=dens, partial=1, frac=frac_res, power=power_res,/all

;Plot oxygen fractional abundances for stage to stage case
set_plot,'ps'
device, /isolatin1, font_index=8
device, bits=8, filename='demo_d_3_ion_unres.ps',         $
        font_size = 14, xsize=18.0, ysize=16.0, $
        yoffset=7.0, /color
device, /helvetica

plot_oo,te,frac_unres.ion[0,*,0],xrange=[1.,1000.],yrange=[1.e-4,1.],$
        title='Oxygen - Stage to stage',$
        xtitle='ELECTRON TEMPERATURE (eV)',$
        ytitle='N!dINDEX!n/N!dTOTAL!n' ,/nodata
for i=0,n_elements(frac_unres.stage)-1 do oplot,te,frac_unres.ion[0,*,i]



device, /close
set_plot,'X'
```

```
!p.font=-1

;Plot oxygen fractional abundances for metastable resolved case
set_plot,'ps'
device, /isolatin1, font_index=8
device, bits=8, filename='demo_d_3_ion_res.ps',         $
        font_size = 14, xsize=18.0, ysize=16.0, $
        yoffset=7.0, /color
device, /helvetica

plot_oo,te,frac_res.ion[0,*,0],xrange=[1.,1000.],yrange=[1.e-8,1.],$
        title='Oxygen - Metastable resolved',$
        xtitle='ELECTRON TEMPERATURE (eV)',$
        ytitle='N!dINDEX!n/N!dTOTAL!n'  ,/nodata
for i=0,n_elements(frac_res.stage)-1 do oplot, te,frac_res.ion[0,*,i]

device, /close
set_plot,'X'
!p.font=-1


;Plot oxygen radiated power for metastable resolved case
set_plot,'ps'
device, /isolatin1, font_index=8
device, bits=8, filename='demo_d_3_power_res.ps',         $
        font_size = 14, xsize=18.0, ysize=16.0, $
        yoffset=7.0, /color
device, /helvetica

plot_oo,te,power_res.plt[0,*],xrange=[1.,1000.],yrange=[1.e-30,1.e-25],$
        title='Oxygen - Metastable resolved',$
        xtitle='ELECTRON TEMPERATURE (eV)',$
        ytitle='POWER FUNCTION (W cm!u3!n)'
oplot,te,power_res.prb[0,*]
oplot,te,power_res.total[0,*] ,thick=3
for i=0,n_elements(power_res.stage)-1 do oplot,te,power_res.ion[0,*,i],line=1

xyouts,0.8,0.63,'total',/normal
xyouts,0.75,0.6,'plt',/normal
xyouts,0.75,0.48,'prb',/normal


device, /close
set_plot,'X'
!p.font=-1


end
```